
Verilog: a brief history

c.1977 First HDLs developed

1985 Verilog introduced by Gateway Design Automation

1990 Gateway purchased by Cadence Design Systems

1991 ‘Opened’ as a language

1995 Made into IEEE standard #1364

2001 Second IEEE standard including extensions and improvements

2005 A new standard with some minor improvements

2005 SystemVerilog introduced: Verilog-2005 plus major extensions ⇒ IEEE #1800

2009 SystemVerilog IEEE Std. 1800-2009 which merges 1364-2005 and 1800-2005

The version used in these modules is Verilog-2001.

[The schematic capture tools used are not compatible with all the features of SystemVerilog.]

Verilog versions

Most of the information in this module is based on Verilog-2001. This is proba-
bly, currently (2010) the most common version although SystemVerilog is gain-
ing ground.

These pages cover some of the significant differences in the major versions.

Verilog-95 is becoming obsolescent. Examples of features absent include:

❏ fewer syntactic options e.g.always @ (a or b or c)
2001 addedalways @ (a, b, c)  andalways @ (*)

❏ No signed registers, shifts etc.

❏ I/O type declarations inside port list
module and2 (a, b, q) // This form for V-95
input a, b;
output q;
...

❏ Multi-dimension arrays

❏ Generate blocks

❏ Parameter passing by name

❏ Most of the file support

This is not an exhaustive list; rather it is intended to be indicative of the type of
enhancements. Verilog-2001 is backwards compatible.

Some of these features may be unfamiliar, at least in Verilog, the first time you
read this. You don’t need to use them, although there will be some brief cover-
age, shortly.

SystemVerilog

SystemVerilog attempts to combine the merits of Verilog HDL with a Hardware
Verification Language (HVL). It is somewhat C++-like.

A selection of additional features include:

❏ Enumerated data types

❏ Multi-dimensional arrays (bundles of buses)

❏ Structures/Unions

❏ Interfaces (related collections of I/O signals)

❏ Object-oriented programming

❏ Assertions

❏ Coverage features

❏ Improved thread synchronization

❏ …

Other ‘existing’ Verilog features have also been enhanced/regularised in places.
Some of these features are for synthesizable RTL code, others are there to
improve the modelling and verification features.

The current standard is 2009; not all toolsets have caught up with this yet.

We don’t use SystemVerilog in the lab. because the Cadence implementation
does not interoperate with some of the other tools we want to use. (As far as we
can make them work!)

VHDL

VHDL is a different HDL which has some popularity. In principle it works
largely in the same way as Verilog. It has some advantages, such as
stronger typing, and disadvantages, such as verbosity. However converting
from one to the other is largely a matter of new syntax.

For further opinions there are numerous VHDL vs Verilog discussions out
there in Internet Land.

Verilog …

❏ … should be, to some extent, already familiar

❏ … is a Hardware Description Language

❍ for modelling hardware

❏ … is designed for parallel programming

❏ … can be synthesized into circuits

❍ providing an appropriate subset is used

❏ … is used as a portable design format

In this module we use Verilog for:

❏ writing tests

❏ implementing hardware

❏ (invisibly) as part of the tool flow

Verilog: a reminder

These pages are provided as a briefaide memoire1. There are many books and
web tutorials which give a more complete description. A suggested, on-line
tutorial may be found at:http://www.asic-world.com/verilog/

This not meant to be comprehensive; you may meet some new material later!

Declarations

Values

Operators
Operators are, in general the same as in C or Java. There are some additional
features:

Assignment

Execution

Control

Timing

Some of these constructs are useful for testbenches but are not synthesizable.1. The IEEE Std 1364-2001 Verilog standard weighs in at 840 pages.

Syntax Use

wire aaa; A simple, combinatorial logic variable.

reg bbb; A variable which may (or may not) be state holding.

reg [7:0] ccc; As above, but an eight bit value.

Syntax Use

1234 A decimal number.

16’hABCD A 16-bit hexadecimal number.

4’b1010 A 4-bit binary number.

x The ‘unknown’ value.

z The high-impedance state.

Syntax Use

&ccc A (unary) reduction operator; ANDs all the bits in the
variable together. Others include ‘|’, ‘^’, ‘~&’, etc.

{ddd, eee} Concatenate two (or more) variables.

{4{ddd}} Concatenate four copies of ddd.

Syntax Use

assign a = b + c; A continuous assignment; combinatorial logic.
Assigns to wire.

d = e & f; Blocking assignment. d may be reassigned in block,
reading top-to-bottom. Use for combinatorial logic.
Assigns to reg.

g <= h; Non-blocking assignment. g should be assigned at most
once at any time. Use for registers. Assigns to reg.

Syntax Use

initial Execute following statement once, starting at time = 0.

always Run following statement continuously. (Needs a delay!)

always @ {…} Run following statement whenever an event in the
parenthesized list occurs.

Syntax Use

if (…) ... {else} ... C/Java like choice.

case Multi-way ‘switch’ choice.

repeat (10) Perform following statement ten times.

while (…) … C-like repetition.

for (…;…;…) C-like repetition.

Syntax Use

#20 Wait for 20 time units.

Verilog reminder: a testbench

initial clk = 0; // Set clock initial value

always #5 clk = ~clk; // Oscillate with 10 unit period

initial #1000 $stop; // Limit simulation run time

initial

begin // Bracket following into one statement

reset = 1; // Initialise the Device Under Test

en = 1; // Set up an initial input value

#10 // Wait one clock cycle

reset = 0; // Remove reset

while (c == 0) #10; // Wait until output ‘c’ asserted

en = 1’x; // Make input ‘en’ undefined

end // End of statement (but time continues)

Verilog for testbenches

This testbench assumes the existence of some design which has defined the var-
iables. It is reminiscent of the stimulus files used in testing in Cadence.

In the example on the slide there are four separate blocks. The three ‘initial’
blocks start at time = 0 and run once. Thus the first statement initialises an input
and stops. The other ‘initial’ statements contain delays which can deschedule
them for periods of (simulation) time.

The always block has no sensitivity list so it runs immediately. A conflict with
the preceding ‘initial’ assignment is avoided by the delay. Following the delay
the clock is inverted and the always runs again; thus the clock oscillates. The
second ‘initial’ block waits for a (long) time and then tells the simulator to halt.
Without this the simulation would run forever – or until it is halted by user inter-
vention.

Note: it would be a mistake to write something like:

always clk = ~clk;

What would happen?

Would the simulation terminate?

The method shown is not the only way to achieve the desired effect.
For example the clock (and termination) may be produced by:

initial
begin
clk = 0;
repeat (200) #5 clk = ~clk;
$stop;
end

You may prefer this style.

Verilog language

Choices

As in most programming languages, there are multiple ways in which a particu-
lar effect can be achieved. The choice of structure will be influenced by the
author’s stylistic prejudices. The best advice is to try and make things as simple
and obvious as you can.

wire p; assign p = !a; // This?
reg q; always @ (a) q = !a; // or this?

Because Verilog syntax has evolved from (arguably) crude beginnings but main-
tained backward compatibility there are sometimes multiple ways to specify the
same thing.

Example
module inv1(input wire a, output wire q);
assign q = !a;
endmodule;

module inv2(a, q);
input a;
output q;
wire a, q;
assign q = ~a;
endmodule;

Both these modules have the same effect.

Notes

❏ These notes are not intended as a comprehensive guide to Verilog.

❏ Introductions to Verilog features may be found in a number of books
and web tutorials.

❍ e.g.http://www.asic-world.com/verilog/

❏ For a definitive guide, consult the IEEE standard document.

Verilog reminder: a circuit

module counter (input wire clk, input wire en,

output reg [3:0] count, output wire c);

reg [3:0] next; // Internal variable (combinatorial)

always @ (count) // Run block when ‘count’ changes

begin

next = count + 1; // Blocking assignment

if (count == 9) next = 0; // Variable may be reassigned

end

always @ (posedge clk, posedge reset)

if (reset) count <= 0; // Asynchronous reset

else if (en) count <= next; // Maybe adopt new value (non-blocking)

else count <= count; // This clause may safely be omitted

assign c = (count == 9); // TRUE is a ‘1’ value

endmodule

Verilog for synthesis

The example is contrived to
illustrate a number of different
Verilog features.

Notes

‘next’ is sometimes reassigned
in its block but it is always
assigned, thus the combinato-
rial block always calculates its
result(s) from its input(s). This
guarantees it is combinatorial and does not need any state holding.

The description of the register, with its asynchronous clear, is one which will be
recognised by the synthesizer.

The register (‘count’) does not need the second else; if it is not enabled it will do
nothing. There is already a state-holding element here so this doesn’t affect the
circuit.

The carry output (‘c’) is assigned combinatorially from a comparator. In practice
this may ‘glitch’ as its input bits change. This doesn’t matter as long as it’s only
sampled on a clock edge.It should not be used as a clock, itself.

It is also possible, and sometimes clearer, to incorporate some of the logic in the
same statement as the register assignment.

always @ (posedge clk, posedge reset)
if (reset) count <= 0; // Asynchronous reset
else if (en)

if (count == 9) count <= 0;
else count <= count + 1;

The choice of style can depend on the author. This is shorter than the example
on the slide although the variable ‘next’ or equivalent – the state that is about to
be entered – is sometimes a convenient input for other logic … or for viewing
for debug purposes.

Guidelines

*** You should read this page! ***

Some ‘features’ of Verilog are open to abuse. It is possible to write Verilog code
in numerous ways. Some ways are error-prone; they ‘look’ right but behave
oddly, or are unduly expensive to build. Some models will run on a simulator.
Some will run on particular simulators, but not on others. Some will run when
synthesised into hardware. Some will be reliable across all tools and implemen-
tations.

Here are some guidelines to help you achieve the last of these cases.

❏ Verilog is case sensitive and ‘wire’ variables do not have to be
declared. Thus:
wire this, that;
assign This = a && b;
assign that = this;
will have an undefined output.

❏ Use blocking assignments for combinatorial logic.

❏ Ensure ‘combinatorial’ outputs are assigned under all conditions.
E.g. include ‘default’ by habit in all ‘case’ statements.

❏ Use non-blocking assignments for latching circuits.

❏ Make variable sizes match. Assigning a variable of one size to one
of another is syntactically legal.

always @ (count)

assign …

always @
(posedge …

next count

c

en

clk

Structural Verilog

Verilog can also be used to specify hierarchical designs.

Example

and2 gate_1 (.a(x), .b(y), .q(z));

where

and2 is the part (module) name

gate_1 is the instance name

{a, b, q} are the inputs/output of the module

{x, y, z} are wires at the higher level

❏ Structural Verilog may be written by hand – but it is often easier to draw schematics

❍ It is sometimes useful for including an existing module in a behavioural description

❏ It is often machine-written as a netlist format

❍ e.g. the output from synthesizing behavioural code may be in this form

❍ As text it is more portable than schematics.

z
a

b
q

x

y

Structural Verilog

Structural Verilog allows the explicit inclusion of certain cells within a design.
The example below shows a simple module with a schematic equivalent.

module demo (input wire clk,
input wire [7:0] a,
input wire b,
input wire c,
output wire p,
output wire [15:0] q);

wire d, e, f;
wire [7:0] w;

assign d = b || c; // Synthesizable statement
block_1 I1(.clk(clk), .a(a), .b(d), .i(e), .j(w));
block_2 I2(.clk(clk), .a(w), .k(f), .l(q));
and2 I3 (.a(e), .b(f), .q(p));
endmodule

Schematic

This example is contrived to show some features:

❏ I/O wires and buses connected directly

❏ Internal wire and bus declarations

❏ Naming of instances

❏ Mixture of structural and synthesized code allowed

❍ Not necessarily Good Practice though!

Syntax

Assume we have a module declaration:

module and2 (input a, input b, output q);
...

The Verilog 2001 syntax

and2 gate_1 (.a(x), .b(y), .q(z));

explicitly connects wires/buses to ports on the module by associating their
names. The order in which these are written is irrelevant. Ports may be omitted
(although this is only sensible for outputs; inputs should always be defined).

Verilog 1995 syntax relies on ordering:

and2 gate_1 (x, y, z);
...

Backwards compatibility means that Verilog 2001 will accept either.

It is suggestedthat the later syntax is preferred because it is more robust against
later edits, e.g. if another port is added to a module.

I3block_1

block_2I1

I2

clk

clk

c

b

[7:0]a

[15:0]q

p

[7:0]w

d
e

f

[7:0]a

[7:0]j

i

b

[7:0]a

[7:0]j

k

b

a

q

Synthesis: (possible) examples

assign y = a & b | c;

assign q = s ? i1 : i0;

always @ (posedge clk, posedge rst)

if (rst) Q <= 0;

else Q <= D;

a

b

c

y

i0

i1

s

s

q

D Q

clk

rst

Slide 6

Synthesis

Verilog is a Hardware Description Language but, if a subset of its features are
used, the description can be turned into an implementation.

Just as a compiler takes a high-level language description of an algorithm and
turns it into machine instructions for a chosen instruction set, a Verilog synthe-
sizer turns a hardware description into an underlying technology.

The technology will vary according to what is on offer on the target process.
Cell libraries are usually available from thesilicon foundry.

A typical cell library will contain familiar gates such as AND2, AND3, OR2, …
[and more - but more on that later].

During synthesis structures are identified in the source – and sometimes this
fails. The bottom example on the slide shows a D-type register with anasyn-
chronous clear. Reassure yourself that you follow the behaviour of this code.

Here’s a genuine example we found with one synthesizer:

always @ (posedge clk, posedge rst)
if (rst || clr) Q <= 0; // Failed to synthesize
else Q <= D;

always @ (posedge clk, posedge rst)
if (rst) Q <= 0;
else if (clr) Q <= 0; else Q <= D;

Behaviourally these blocks are the same. They should simulate in the same way.
However the author needed to separate the asynchronous input explicitly for
synthesis.

FPGA …

When synthesizing for an FPGA the tool will look for structures it can recognise
and try to map these into the available resources.

Examples:

❏ A RAM – if specified appropriately – may be use dedicated RAM
blocks.

❏ A statement such as “A = B + C;” will use adjacent LUTs for its
various bits to exploit fast carry logic.

❏ An “ initial ” statement may be
applied to a registered signal. All flip-
flops are zeroed when the FPGA’s are
configured. “initial x = 1; ” can
be implemented by inverting the value around the actual flip-flop.

❏ An FPGA with multiplier blocks may be able to exploit these.

Logic functions will
then be amalgamated to
fit into LUTs of the
appropriate size.

If a statement cannot be mapped into an available
structure then the design may not fit the FPGA.

On the right are two RAM read structures. The upper
one does not map to Xilinx block RAMs and therefore
the RAM will be expanded into (a LOT of) combinato-
rial logic.

The lower structurelooksbigger but exploits the avail-
able resources and thus is (much) more efficient.

C compiler Verilog

Syntax parsing Syntax parsing

Semantic Analysis ‘Elaboration’

Optimisation Optimisation

Code dump Technology mapping

x

Starts as ‘0’

a

b

s

q

as b q
00 0 0

?
x

00 1 0x
10 0 1x
10 1 1x
01 0 0x
01 1 1x
11 0 0x
11 1 1x

RAM

RAM

RAM

RAM

States and truth tables

Verilog represents circuits digitally.

❏ There are four basic states which a digital wire can be in: {0, 1, Z, X}

❏ Switching between states is instantaneous

❏ There is no concept of ‘halfway’ or of ‘edge speed’

❏ ‘X’ (unknown) usually means either 0 or 1 but the simulator doesn’t know which

❍ This is the default initial state of flip-flops, latches etc.

❍ It may occur if two different outputs are connected together (mistake!)

❏ ‘Z’ (high-impedance, “Tristate™”) means there is no output driving the network

❍ No longer relevant for on-chip signals

❍ Default state for input wires

0 1 X Z
0 1 X X0

XOR

1 0 X X1
X X X XX
X X X XZ

0 1 X Z
0 0 0 00

AND

0 1 X X1
0 X X XX
0 X X XZ

0 1 X Z
0 1 X X0

OR

1 1 1 11
X 1 X XX
X 1 X XZ

Tasks/Functions

A task is a procedure/method/subroutine/… structure in Verilog. They are local
to modules (but could be ‘included’). Tasks may substitute for repetitive opera-
tions and may be useful in simplifying and clarifying a test bench. They may
contain combinatorial or sequential logic and occupy (simulation) time in exe-
cution. [Syntax may vary; an alternative example is given in the lab. manual.]

task something;
input …
output …
begin
<behavioural code>
end
endtask

Functions are like simple tasks returning a value. They are typically used to
represent some combinatorial function. For example:

function adder; // Function name is also result
input x, y;
begin
adder = x + y; // Hardly worthwhile!
end
endfunction

...

assign sum = adder(a, b);

Compiler directives

Verilog complier directives are keywords preceded by the backtick character‘

Some of the more common ones are:

‘timescale

Set the timestep in simulation, i.e. how long#1  really represents.

‘include

Import another file into the source; useful for including common definitions
amongst several files in a project.

‘define

Verilog allows test substitution in a similar manner to C’s “#define”.

‘define THING 99 // Define text substitution
...
x <= ‘THING; // In use (note ‘)

It is recommended to be used in the same way, and for the same reasons.

‘undef

Remove a compile-time definition.

‘ifdef, ‘else, ‘elsif, ‘endif, ‘ifndef

Used to control conditional compilation by ‘bracketing’ parts of the code. The
application should be obvious.

Generate

module big_memory (input wire CE, input wire WE, input wire OE,
input wire A[11:0], input wire [7:0] Din,
output wire [7:0] Dout);

reg  [3:0] En;
wire [7:0] Data [0:3] // Array of four data buses

always @ (CE, A[11:10]) // Address decoder with enable
begin
En = 4’b0000;
if (CE) En[A[11:10]] = 1’b1;
end

generate
genvar i; // Variable used for compile-time iteration
for (i=0; i<4; i=i+1)

begin: block
memory mem((.CE(En[i]), .WE(WE), .OE(OE),

.A(A[9:0]), .Din(Din), .Dout(Data[i]));
end

endgenerate

assign Dout = Data[A[11:10]]; // Output multiplexer

end

Generate

Generate allows multiple instances to be placed iteratively. The example in the
slide shows how a 4 KB memory could be constructed from four, 1 KB blocks.

❏ Thealways  block provides a 2:4 decoder from the upper two
address lines.

❍ a 2-bit field taken from the address is used to index anEn bit,
which is set to ‘1’; all the (other) bits have been zeroed.

❏ The iterated instantiation is delimited by {generate , endgen-
erate }

❏ genvar  declares a control variable; this only exists at ‘compile
time’

❏ thefor  loop iterates using the control variable

❏ instances of memory called ‘mem[0:3] ’ are produced

❏ wiring is in common except for the different in chip selects
(En[0:3] )

❏ as an illustration, the output data buses are multiplexed explicitly.
(OE serves no real purpose here.)

This example instantiation iterates a structural Verilog block: it is possible to
contain a variety of statements within the block.

The control for the generate needs to be determined when the blocks are to be
instantiated. Hardware cannot be created dynamically ‘at run time’. Obviously!

The real value of such constructs is apparent when the number of instantiations
is parameterised. For example:

...

always @ (CE, A[9+‘BITS:10])
begin
En = 0;
if (CE) En[A[9+‘BITS:10]] = 1’b1;
end

generate
genvar i;
for (i=0; i<(1<<(‘BITS-1)); i=i+1)

...

This allows the iterations to be specified with a single input definition.

Here ‘BITS specifies the log2 of the number of blocks and thus1<<‘BITS

regenerates this number. It should be constrained to 1 because “A[10:10]” is
legal but “A[9:10]” will cause problems.

The reverse approach is more difficult as taking a log is more tedious. However
here is an appropriate function from the Verilog Standard document.

function integer clogb2;
input [31:0] value;
for (clogb2=0; value>0; clogb2=clogb2+1;)

value = value>>1;
endfunction

If the log2 is non-integral, this finds the next largest integer, which is generally
what is wanted.

CE

WE

OE

Din

A

Din

A

Din

A

Din

DoutDoutDoutDout

CE

WE

OE

CE

WE

OE

CE

WE

OE

Data[0]

mem[0] mem[1] mem[2] mem[3]

Data[1]
Data[2] Data[3]

Din
A

A

OE

WE

CE

Dout

En[0]

En[3]
En[2]

En[1]

Parameters

module adder (a, b, s);
parameter n = 16;
input wire [n-1:0] a, b;
output wire [n-1:0] s;

assign s = a + b;

endmodule

module example;
wire [15:0] pp, qq, rr;
wire [31:0] ss, tt, uu;
wire [7:0] vv, ww, xx;

adder add16(pp, qq, rr);
adder add32 #(32) (ss, tt, uu); // Verilog 95
adder add8 #(.n(8)) (.a(vv), .b(ww), .c(xx)); // Verilog 2001

endmodule

Parameters

There are several ways to define and modify parameters in Verilog. Only a selec-
tion is elucidated here but there should be enough to illustrate the concept.

The example on the slide specifies a (somewhat redundant) module and instanti-
ates it three times. The module has a single parameter (n) which, in this case,
specifies its bus widths. There is a ‘default’ value of 16 assigned to this parame-
ter.

The first instantiation simply uses this default value. The other instantiations
override the default. For clarity, in the first instantiation it is suggested that a
‘16’ should be passed anyway.

The add32 uses the older Verilog-95 syntax both the parameter list and the con-
nections. This syntax works much like many programming languages in that
associations are determined by the order of the lists. this syntax is also useable
in later versions of Verilog.

add8 is instantiated using Verilog 2001 syntax. This specifies each parameter/
connection explicitly.

.<name_inside_instance>(<local_name>)

In this syntax the elements can be specified in any order.

It is recommended that the Verilog 2001 syntax is used. It is more verbose but
less prone to errors, especially if lists are edited after creation. In the case of
parameters it also allows an arbitrary set of overrides to be used, not just those
from the start of the list.

defparam

A parameter can be passed explicitly to a module using ‘defparam ’.

The following could be added to the example above:

defparam add16.n = 16;

where a value is assigned using its hierarchical name.

Summary

❏ Verilog is a Hardware Description Language

❍ Quite good for modelling

❍ Highly parallel

❍ A subset can be used as source for RTL synthesis

❏ Verilog has evolved

❍ Not always syntactically ‘clean’

❍ Multiple ways to write the same thing

❏ There are ‘traps’ for the unwary

❍ Synthesizers look for structures to implement

❍ Simulated and synthesized functions may not match (if language ‘abused’)

❍ Helpful to keep implementation technology in mind.

❍ Care needed not to produce redundant hardware (e.g. latches)

❏ Useful … with a degree of care

